
UNION  
این تابع سطرهای دو یا چند جدول را برداشته و همه را در یک جدول واحد زیر هم قرار می‌دهد.
ساختار (Syntax)
dax
UNION(<table>, <table>, [<table>], ...)
· table: جدول‌هایی که می‌خواهید به هم متصل کنید.
· خروجی: یک جدول واحد شامل تمام سطرهای جدول‌های ورودی.

قوانین طلایی UNION
۱. تعداد ستون‌ها: تمام جدول‌ها باید تعداد ستون‌های برابری داشته باشند.
۲. ترتیب ستون‌ها: ستون‌ها بر اساس موقعیت (Position) زیر هم قرار می‌گیرند، نه بر اساس نام. (یعنی ستون اولِ جدول اول زیر ستون اولِ جدول دوم می‌رود).
۳. تکرار: این تابع سطرهای تکراری را حذف نمی‌کند (برخلاف برخی دستورات SQL). اگر یک سطر در هر دو جدول باشد، در خروجی ۲ بار ظاهر می‌شود.

مسئله: مدیریت از ما می‌خواهد بدانیم در کدام «بازه زمانی ۲ ساعته» از ساعات کاری، بیشترین تعداد سفارش کار (Work Order) ثبت شده است. چالش اصلی اینجاست که ما در مدل داده‌ای خود، جدولی برای «بازه‌های ۲ ساعته» نداریم و باید آن را به صورت مجازی در فرمول بسازیم و سپس داده‌ها را بر اساس آن دسته‌بندی کنیم.
اگه میخواستم با کوئری sql  حل کنم:
WITH TimeSlots AS (
    SELECT '08-10' AS Label, 8 AS StartH, 10 AS EndH UNION ALL
    SELECT '10-12', 10, 12 UNION ALL
    SELECT '12-14', 12, 14 UNION ALL
    SELECT '14-16', 14, 16 UNION ALL
    SELECT '16-18', 16, 18 UNION ALL
    SELECT '18-20', 18, 20
),
SlotCounts AS (
    SELECT
        ts.Label,
        COUNT(wo.Id) AS WO_Count
    FROM TimeSlots ts
    LEFT JOIN mfr.WorkOrder wo
        ON DATEPART(HOUR, wo.CreatedOn) >= ts.StartH
       AND DATEPART(HOUR, wo.CreatedOn) <  ts.EndH
    GROUP BY ts.Label
),
MaxSlot AS (
    SELECT MAX(WO_Count) AS MaxCount
    FROM SlotCounts
)
SELECT sc.Label
FROM SlotCounts sc
JOIN MaxSlot ms
    ON sc.WO_Count = ms.MaxCount;
گام اول: ساخت جدول مجازی (Virtual Table)
ابتدا باید یک جدول حاوی نام بازه، ساعت شروع و ساعت پایان بسازیم. از توابع UNION (برای چسباندن سطرها) و ROW (برای ایجاد هر سطر) استفاده می‌کنیم.
EVALUATE 
    UNION(
        ROW("Label", "08-10", "StartH", 8,  "EndH", 10),
        ROW("Label", "10-12", "StartH", 10, "EndH", 12),
        ROW("Label", "12-14", "StartH", 12, "EndH", 14),
        ROW("Label", "14-16", "StartH", 14, "EndH", 16),
        ROW("Label", "16-18", "StartH", 16, "EndH", 18),
        ROW("Label", "18-20", "StartH", 18, "EndH", 20)
    )
[image: ]
گام دوم: شمارش ردیف‌به‌ردیف (Iteration)
با استفاده از ADDCOLUMNS و COUNTX به سراغ هر بازه می‌رویم. در هر سطر، کل جدول سفارشات را فیلتر می‌کنیم تا فقط سفارشاتی که در آن ساعت خاص ثبت شده‌اند شمارش شوند.
EVALUATE
VAR TimeSlots =
    UNION(
        ROW("Label", "08-10", "StartH", 8,  "EndH", 10),
        ROW("Label", "10-12", "StartH", 10, "EndH", 12),
        ROW("Label", "12-14", "StartH", 12, "EndH", 14),
        ROW("Label", "14-16", "StartH", 14, "EndH", 16),
        ROW("Label", "16-18", "StartH", 16, "EndH", 18),
        ROW("Label", "18-20", "StartH", 18, "EndH", 20)
    )
VAR SlotWithCount =
    ADDCOLUMNS(
        TimeSlots,
        "WO_Count",
            COUNTX(
                FILTER(
                    ALL(workorder),
                    HOUR(WorkOrder[CreatedOn]) >= [StartH]
                        && HOUR(WorkOrder[CreatedOn]) < [EndH]
                ),
                1
            )
    )
RETURN
    SlotWithCount



گام سوم: استخراج نتیجه نهایی
بزرگترین عدد به دست آمده را پیدا کرده و نام (Label) مربوط به آن را نمایش می‌دهیم.
Top 2H Time Slot = 
-- مرحله ۱: تعریف بازه‌های زمانی (ساخت جدول پایه)
VAR TimeSlots =
    UNION(
        ROW("Label", "08-10", "StartH", 8,  "EndH", 10),
        ROW("Label", "10-12", "StartH", 10, "EndH", 12),
        ROW("Label", "12-14", "StartH", 12, "EndH", 14),
        ROW("Label", "14-16", "StartH", 14, "EndH", 16),
        ROW("Label", "16-18", "StartH", 16, "EndH", 18),
        ROW("Label", "18-20", "StartH", 18, "EndH", 20)
    )

-- مرحله ۲: محاسبه تعداد سفارشات برای هر سطر جدول بالا
VAR SlotWithCount =
    ADDCOLUMNS(
        TimeSlots,
        "WO_Count",
            COUNTX(
                FILTER(
                    ALL(WorkOrder), -- نادیده گرفتن فیلترهای مزاحم برای دیدن کل سفارشات
                    HOUR(WorkOrder[CreatedOn]) >= [StartH]
                        && HOUR(WorkOrder[CreatedOn]) < [EndH]
                ),
                1
            )
    )

-- مرحله ۳: پیدا کردن بالاترین تعداد ثبت شده
VAR MaxCount =
    MAXX(SlotWithCount, [WO_Count])

-- مرحله ۴: برگرداندن نامِ شلوغ‌ترین بازه
RETURN
    MAXX(
        FILTER(SlotWithCount, [WO_Count] = MaxCount),
        [Label]
    )




۴. نکات کلیدی 
· چرا از ALL استفاده کردیم؟
در مرحله ۲، اگر از ALL استفاده نکنیم، فرمول فقط سفارشاتِ مربوط به همان روز یا فیلتری که کاربر انتخاب کرده را می‌بیند. ALL اجازه می‌دهد محاسبات ما روی تمامِ بانک اطلاعاتی انجام شود.
· قدرت ADDCOLUMNS:
این تابع به ما اجازه می‌دهد روی یک جدول مجازی که در حافظه ساخته‌ایم، محاسبات جدید انجام دهیم و نتایج را در ستون‌های موقت ذخیره کنیم.
تمرین: سوال قبل انگاری ساب کوئری نوشتیم. مگه نه؟ خب نسخه بهینه تر بنویسید. (5 نمره)
تمرین: syntax توابعی که در این مسئله یاد گرفتید بنویسید. (نمره نداره کاری نداره آخه)
CONCATENATEX 
تابع CONCATENATEX عضوی از خانواده توابع Iterator (تکرارکننده) است (مثل SUMX یا MAXX).
کار این تابع این است: سطر به سطرِ یک جدول را می‌خواند، یک عبارت (متن) را برای هر سطر محاسبه می‌کند و در نهایت تمام آن متن‌ها را با یک «جداکننده» (مثل ویرگول) پشت سر هم می‌چسباند.
ساختار (Syntax)
CONCATENATEX(<table>, <expression>, [delimiter], [orderBy_expression], [order])
1. table: جدولی که می‌خواهیم روی سطرهایش حرکت کنیم.
2. expression: چه چیزی از هر سطر برداشته شود؟ (مثلاً نام ماه).
3. delimiter: متن‌ها با چه علامتی از هم جدا شوند؟ (مثلاً " - " یا "، ").
4. orderBy_expression (اختیاری): بر چه اساسی مرتب شوند؟
5. order (اختیاری): صعودی یا نزولی؟

مثال ۱: لیست کردن ماه‌های انتخاب شده
فرض کن کاربر در اسلایسر چند ماه را انتخاب کرده و تو می‌خواهی در یک کارت بنویسی: «ماه‌های انتخاب شده: فروردین، اردیبهشت».
SelectedMonthsList = 
VAR __SelectedTable = ALLSELECTED('master DimDate'[PersianMonthName]) 
RETURN
CONCATENATEX(
    __SelectedTable, 
    'master DimDate'[PersianMonthName], 
    " - "
)

اگر کاربر فروردین، تیر و آذر را انتخاب کرده باشد، خروجی می‌شود: فروردین - تیر - آذر

تفاوت CONCATENATE با CONCATENATEX
· CONCATENATE: فقط دو تا متن ساده را به هم می‌چسباند.
· CONCATENATEX: روی یک جدول می‌چرخد و مقادیر چندین سطر را به هم می‌چسباند

ترکیب با توابع قبلی (یک مثال حرفه‌ای)
بیا با HASONEVALUE ترکیبش کنیم. اگر یک ماه انتخاب شده بود، همان را نشان بده، اگر بیشتر بود لیستشان کن:
DynamicTitle = 
IF(
    HASONEVALUE('master DimDate'[PersianMonthName]),
    "گزارش ماه: " & SELECTEDVALUE('master DimDate'[PersianMonthName]),
    "گزارش ماه‌های: " & CONCATENATEX(VALUES('master DimDate'[PersianMonthName]), 'master DimDate'[PersianMonthName], " و ")
)

نفس عمیق بکشید. بریم مسئله بعدی:
 مسئله: پیدا کردن بازه‌ای که بیشترین تنوع نیروی انسانی (Owner) را دارد
WITH TimeSlots AS (
    SELECT '08-10' AS Label, 8 AS StartH, 10 AS EndH UNION ALL
    SELECT '10-12', 10, 12 UNION ALL
    SELECT '12-14', 12, 14 UNION ALL
    SELECT '14-16', 14, 16 UNION ALL
    SELECT '16-18', 16, 18 UNION ALL
    SELECT '18-20', 18, 20
)
    SELECT
        ts.Label,
        COUNT(DISTINCT wo.OwnerId) AS WO_Count
    FROM TimeSlots ts
    LEFT JOIN mfr.WorkOrder wo
        ON DATEPART(HOUR, wo.CreatedOn) >= ts.StartH
       AND DATEPART(HOUR, wo.CreatedOn) <  ts.EndH
	   WHERE wo.CompanyId = 268
      AND FORMAT(wo.CreatedOn, 'yyyy', 'fa') = 1404 --AND wo.RevisionId IS null
    GROUP BY ts.Label

راه حل با DAX
Top Slot by Staff Diversity = 
-- مرحله ۱: تعریف همان بازه‌های زمانی ۲ ساعته
VAR TimeSlots =
    UNION(
        ROW("Label", "08-10", "StartH", 8,  "EndH", 10),
        ROW("Label", "10-12", "StartH", 10, "EndH", 12),
        ROW("Label", "12-14", "StartH", 12, "EndH", 14),
        ROW("Label", "14-16", "StartH", 14, "EndH", 16),
        ROW("Label", "16-18", "StartH", 16, "EndH", 18),
        ROW("Label", "18-20", "StartH", 18, "EndH", 20)
    )

-- مرحله ۲: محاسبه تعداد "کارمندان منحصربه‌فرد" در هر بازه
VAR SlotWithStaffCount =
    ADDCOLUMNS(
        TimeSlots,
        "Unique_Owners",
            -- ابتدا کل سفارش‌های آن بازه را فیلتر می‌کنیم
            VAR __OrdersInSlot = 
                FILTER(
                    ALL(WorkOrder),
                    HOUR(WorkOrder[CreatedOn]) >= [StartH]
                        && HOUR(WorkOrder[CreatedOn]) < [EndH]
                )
            RETURN
               
                    COUNTROWS(
                        DISTINCT(
                            SELECTCOLUMNS(
                                __OrdersInSlot,
                                "OwnerId", WorkOrder[OwnerId]
                            )
                        )
                    )
    )

-- مرحله ۳: پیدا کردن بیشترین تعداد کارمند ثبت شده در یک بازه
VAR MaxStaff = MAXX(SlotWithStaffCount, [Unique_Owners])

-- مرحله ۴: برگرداندن برچسب آن بازه
RETURN
    MAXX(
        FILTER(SlotWithStaffCount, [Unique_Owners] = MaxStaff),
        [Label]
    )

تفاوت کلیدی  VALUES در مقابل ALL
ALL  به فیلترها نگاه نمی‌کند و همه مقادیر کل تاریخچه را می‌آورد.
· VALUES: به فیلترهای کاربر (Slicer) احترام می‌گذارد و فقط مقادیر موجود در فیلتر فعلی را می‌آورد.
تابع DISTINCT برادر دوقلوی VALUES است، اما یک تفاوت کوچک و حیاتی در رفتار با «داده‌های ناموجود» دارد.
تابع DISTINCT هم مثل VALUES عمل می‌کند؛ یعنی به یک ستون نگاه کرده و مقادیر تکراری را حذف می‌کند تا یک لیست از مقادیر یکتا بسازد.
اما یک تفاوت طلایی وجود دارد: اگر در مدل داده‌ای شما ناهماهنگی وجود داشته باشد (مثلاً در جدول فروش، محصولی ثبت شده باشد که در جدول محصولات وجود ندارد)، تابع VALUES یک ردیفِ خالی (Blank) به لیست اضافه می‌کند تا آن خطا را نشان دهد، اما DISTINCT فقط مقادیری که واقعاً وجود دارند را می‌آورد و با ردیف‌های خالیِ ناشی از ناهماهنگی کاری ندارد.
در Power BI، ناهماهنگی یا رکوردهای یتیم به چند دلیل رایج اتفاق می‌افتند:
۱. بارگذاری داده‌های کثیف (Data Silos)
در بسیاری از سازمان‌ها، داده‌ها از منابع مختلف می‌آیند. مثلاً:
· جدول فروش از سیستم CRM می‌آید.
· جدول محصولات از یک فایل Excel می‌آید.
وقتی این دو را در Power BI به هم وصل می‌کنید، Power BI مثل SQL سخت‌گیری نمی‌کند که مانع ورود داده شود؛ بلکه رابطه را می‌سازد اما خودش یک ردیف مجازی به نام Unknown Member (همان Blank) در سمت جدول Dimension ایجاد می‌کند تا تراکنش‌های یتیمِ جدول Fact بدون صاحب نمانند.
۲. زمان‌بندی به‌روزرسانی (Refresh Timing)
فرض کن امروز یک محصول جدید فروخته شده و در جدول Fact ثبت شده است. اما ادمین سیستم هنوز محصول جدید را در جدول مرجع (Dimension) تعریف نکرده یا کوئریِ مربوط به محصولات هنوز ریفرش نشده است. در این لحظه، مدل شما دچار ناهماهنگی می‌شود.
۳. فیلترهای دوطرفه (Bi-directional Filters)
گاهی اوقات ناهماهنگی به خاطر نبودنِ فیزیکی داده نیست، بلکه به خاطر فیلترهایی است که باعث می‌شوند یک رابطه در لحظه محاسبات، مقداری را در سمت Dimension پیدا نکند.
خونسردی خودتون رو حفظ کنید بریم مسئله بعدی:

تابع ISINSCOPE
 این تابع ابزاری است برای اینکه بفهمیم «الان در کدام سطح از سلسله‌مراتب (Hierarchy) هستیم؟»
این تابع زمانی کاربرد دارد که شما از فیلدهای تاریخ (مثل سال، فصل، ماه) در سطرهای یک ماتریس یا نمودار استفاده می‌کنید و می‌خواهید محاسبات در هر سطح متفاوت باشد.
ساختار
ISINSCOPE(<columnName>)
خروجی این تابع TRUE یا FALSE است.
تفاوت مهم با HASONEVALUE
بسیاری از کاربران این دو را اشتباه می‌گیرند:
· HASONEVALUE: چک می‌کند که آیا کلاً یک مقدار در فیلتر باقی مانده یا نه (تحت تأثیر اسلایسر هم هست).
· ISINSCOPE: چک می‌کند که آیا این ستون عامل اصلیِ ایجاد آن سطر در گزارش هست یا نه (مخصوص تحلیل سلسله‌مراتب در ماتریس).

مسئله:
فرض کن ماتریسی داری که در سطرهای آن ابتدا PersianCalendarYear (سال) و زیر آن PersianMonthName (ماه) قرار دارد.
می‌خواهیم فرمولی بنویسیم که:
1. اگر در سطح ماه بودیم، سطح دو را نشان دهد.
2. اگر در سطح سال بودیم، بنویسد "سطح یک".
ScopeTest = 
SWITCH(
    TRUE(),
    ISINSCOPE('master DimDate'[PersianMonthName]), "سطح دو",
    ISINSCOPE('master DimDate'[PersianCalendarYear]), "سطح یک",
   "هیچی"
)

شرط‌ها را از ریزترین سطح به درشت‌ترین سطح بنویس

کاربرد طلایی: محاسبه درصد از والد (Percent of Parent)
یکی از سخت‌ترین کارها در دکس، محاسبه سهم یک ماه از «همان سال» است (نه از کل تاریخ). با ISINSCOPE و ALLSELECTED این کار ساده می‌شود:
PercentOfParent = 
VAR __CurrentSales = SUM(Sales[Amount])
RETURN
SWITCH(
    TRUE(),
    -- اگر در سطح ماه هستیم، باید تقسیم بر کل فروش "آن سال" کنیم
    ISINSCOPE('master DimDate'[PersianMonthName]), 
        VAR __YearSalesTable = ALLSELECTED('master DimDate'[PersianMonthName])
        VAR __YearSales = SUMX(__YearSalesTable, SUM(Sales[Amount]))
        RETURN __CurrentSales / __YearSales,
    
    -- اگر در سطح سال هستیم، سهم سال از کل را نشان بده
    ISINSCOPE('master DimDate'[PersianCalendarYear]),
        VAR __AllSalesTable = ALLSELECTED('master DimDate')
        VAR __AllSales = SUMX(__AllSalesTable, SUM(Sales[Amount]))
        RETURN __CurrentSales / __AllSales,
        
    BLANK()
)

چرا در 2025 از ISINSCOPE استفاده می‌کنیم؟
در گزارش‌های مدرن که کاربران زیاد از قابلیت Drill-down (ریزه شدن در داده‌ها) استفاده می‌کنند، این تابع به شما اجازه می‌دهد:
1. ظاهر گزارش را تمیز کنید: (مثلاً در ردیف‌های جمع کل، اعداد بی‌معنی را نمایش ندهید).
2. محاسبات داینامیک: فرمول در سطح «فصل» یک رفتار داشته باشد و در سطح «روز» رفتاری دیگر.
خلاصه برای یادآوری:
· فکر کن ISINSCOPE مثل یک GPS عمل می‌کند که به کدِ تو می‌گوید: «الان دقیقاً توی کدوم طبقه (سال، فصل یا ماه) ایستادی؟»

تابع CONTAINSROW ابزاری است برای «چک کردن وجودِ مقادیر». این تابع بررسی می‌کند که آیا یک سطرِ خاص (یا مجموعه‌ای از مقادیر) در یک جدول وجود دارد یا خیر.
در واقع، این تابع نسخه بهینه و تمیزِ عملگر IN است.
ساختار (Syntax)
dax
CONTAINSROW(<table>, <value1>, <value2>, ...)
· table: جدولی که می‌خواهیم در آن جستجو کنیم.
· values: مقادیری که دنبالشان می‌گردیم (باید به همان تعداد و ترتیب ستون‌های جدول باشند).
· خروجی: یک مقدار منطقی (TRUE یا FALSE).

مثال ۱: چک کردن انتخاب‌های کاربر (با DimDate)
فرض کن می‌خواهی چک کنی که آیا ماه «فروردین» جزو ماه‌های انتخاب شده توسط کاربر هست یا نه، تا بر اساس آن یک پیام خاص نشان دهی.
IsFarvardinSelected = 
VAR __SelectedMonths = VALUES('master DimDate'[PersianMonthName]) -- جدولی از ماه‌های انتخاب شده
RETURN
CONTAINSROW(__SelectedMonths, "فروردین")


اگر فروردین در لیست انتخاب‌های کاربر باشد، خروجی TRUE می‌شود.


تفاوت CONTAINSROW با عملگر IN
در واقع این دو معادل هم هستند:
1. PersianMonthName IN {"فروردین", "تیر"}
2. CONTAINSROW({"فروردین", "تیر"}, PersianMonthName)
اما طبق فلسفه DAX for Humans، استفاده از CONTAINSROW خوانایی کد را بالا می‌برد چون دقیقاً مشخص است که دارید عملیات جستجو را روی یک Row (سطر) انجام می‌دهید.

مثال ۳: جستجوی ترکیبی (چند ستونه)
این تابع زمانی قدرتمندتر می‌شود که بخواهی ترکیب دو ستون را چک کنی. مثلاً چک کردن یک سال و یک ماه خاص:
IsSpecificDateSelected = 
VAR __SelectedData = VALUES(DimDate) 
RETURN
CONTAINSROW(
    SELECTCOLUMNS(__SelectedData, "Year", DimDate[PersianCalendarYear], "Month", DimDate[PersianMonthName]),
    1403, 
    "فروردین"
)
این فرمول چک می‌کند که آیا در جدولِ فیلتر شده توسط کاربر، سطری وجود دارد که سالش ۱۴۰۳ و ماهش فروردین باشد؟
خلاصه برای ذهن شما:
· ورودی اول: یک جدول (یا لیستی داخل { }).
· ورودی‌های بعدی: مقادیری که می‌خواهیم پیدا کنیم.
· خروجی: فقط TRUE یا FALSE.
· کاربرد اصلی: ساده‌سازی شرط‌های پیچیده و چک کردن انتخاب‌های کاربر در اسلایسر.
نکته برای سال ۲۰۲۵: در نسخه‌های جدید Power BI، استفاده از این تابع برای مدیریت فیلترهای پویا در محاسبات سنگین، از نظر پرفورمنس بسیار توصیه می‌شود.
مطابق با متدولوژی DAX for Humans، تابع EXCEPT یکی از توابع «جدولی» (Table Functions) است که برای مقایسه دو جدول و پیدا کردن تفاوت‌های آن‌ها به کار می‌رود.
به زبان ساده: EXCEPT تمام سطرهایی را که در «جدول اول» هست ولی در «جدول دوم» نیست، برمی‌گرداند. (عملیات تفریق دو مجموعه).
ساختار (Syntax)
EXCEPT(<left_table>, <right_table>)
· left_table: جدول پایه (هر چه اینجا باشد و در جدول دوم نباشد، استخراج می‌شود).
· right_table: جدولی که می‌خواهیم مقادیرش را از جدول اول حذف کنیم.

نکته بسیار مهم (قانون طلایی)
برای اینکه این تابع کار کند، هر دو جدول باید تعداد ستون‌های برابر و ترتیب دیتاتایپ‌های یکسان داشته باشند.

مثال ۱: پیدا کردن ماه‌هایی که فروش نداشته‌اند
فرض کن می‌خواهی لیست ماه‌هایی از سال ۱۴۰۳ را پیدا کنی که هیچ فروشی در آن‌ها ثبت نشده است.
dax
MonthsWithNoSales = 
VAR __AllMonths = VALUES(DimDate[PersianMonthName]) -- لیست تمام ۱۲ ماه
VAR __MonthsWithSales = VALUES(Sales[MonthName]) -- لیست ماه‌هایی که در جدول فروش هستند
RETURN
EXCEPT(__AllMonths, __MonthsWithSales)
خروجی این تابع یک جدول است که فقط نام ماه‌های بدون فروش را در خود دارد.

مثال ۲: ترکیب با CONCATENATEX برای گزارش‌دهی
می‌توانیم از مثالی که یاد گرفتی استفاده کنی تا یک پیغام پویا بسازی:
dax
MissingMonthsAlert = 
VAR __AllMonths = VALUES(DimDate[PersianMonthName])
VAR __MonthsWithSales = VALUES(Sales[MonthName])
VAR __MissingTable = EXCEPT(__AllMonths, __MonthsWithSales)
RETURN
"ماه‌های بدون فروش: " & CONCATENATEX(__MissingTable, [PersianMonthName], "، ")

مثال ۳: مقایسه دو سال (مثلاً ۱۴۰۲ نسبت به ۱۴۰۳)
فرض کن می‌خواهی ببینی کدام شعبه‌ها (BranchId) در سال ۱۴۰۲ فعال بوده‌اند ولی در سال ۱۴۰۳ هیچ فعالیتی نداشته‌اند.
dax
InactiveBranches = 
VAR __Branches1402 = CALCULATETABLE(VALUES(DimDate[BranchId]), DimDate[PersianCalendarYear] = 1402)
VAR __Branches1403 = CALCULATETABLE(VALUES(DimDate[BranchId]), DimDate[PersianCalendarYear] = 1403)
RETURN
EXCEPT(__Branches1402, __Branches1403)
(نکته: چون هنوز CALCULATETABLE را به صورت رسمی در متدولوژی خودمان نگفتیم، فرض کن دو لیست از شعبه‌ها داری).

کاربرد در عیب‌یابی (Debugging)
یکی از بهترین کاربردهای EXCEPT در سال ۲۰۲۵ برای دکس‌نویس‌ها، پیدا کردن دیتاهای ناقص است. مثلاً:
· پیدا کردن مشتریانی که ثبت‌نام کرده‌اند اما خرید نکرده‌اند.
· پیدا کردن کالاهایی که در انبار هستند ولی در لیست فروش نیستند.
خلاصه برای ذهن شما:
· EXCEPT یعنی: «این‌ها رو داشته باش، ولی اون‌هایی که توی لیست دوم هست رو ازشون حذف کن».
· خروجی آن همیشه یک Table است.
· ترتیب نوشتن جدول‌ها مهم است (جدول اول منهای جدول دوم).
تمرین: فکر کن چطور می‌توانی با COUNTROWS و EXCEPT تعداد ماه‌هایی که هنوز در آن‌ها فروشی انجام نشده را بشماری؟
(پاسخ: COUNTROWS(EXCEPT(AllMonths, MonthsWithSales)))
مطابق با متدولوژی DAX for Humans، تابع UNION برای «چسباندن جدول‌ها روی هم» استفاده می‌شود. اگر EXCEPT عمل تفریق بود، 

